



DECUS

PROGRAM LIBRARY

DECUS NO.	8-359
TITLE	HI-Q GAME PLAYING PROGRAM
AUTHOR	M. L. Fichtenbaum and R. E. Peterson
COMPANY	General Radio Company Concord, Massachusetts
DATE	September 22, 1970
SOURCE LANGUAGE	General Radio's PDP-8/1130 Assembly (Similar to PAL III)

HI-Q GAME PLAYING PROGRAM

DECUS Program Library Write-up

DECUS NO. 8-359

HI-Q is a game played with pegs on a cross-shaped board. The board contains 33 holes, symmetrically arranged. Initially, all holes except the one in the center contain removable pegs.

A move is made by "jumping" a peg over an adjacent peg into a vacant hole, then removing the peg which was jumped. A move can be made only when two pegs and a hole lie in a vertical or horizontal line. A game consists of a series of moves, starting from the initial configuration of the board, and ending when no more moves can be made. A complete, winning game results in only one peg remaining, in the center position.

This program employs the PDP-8 to find winning-game solutions to HI-Q by an exhaustive search method. At each move, a "tree" is formed of all possible moves, and the first move found is made. Whenever a situation exists where no further moves can be made, the program backs up one step along the tree and makes the next move. Any winning solution is printed as a series of 32 board images. Other printouts are available by means of the SWITCH REGISTER setting.

Program operation

The program begins at location 200. The switch functions available are:

- ```

0 print all 32-move games (if off, print only those
 with peg in center hole)
1 print all games
2 print octal move counts for all games
10 print only serial number for each printed game
 (if off, print serial number and board images)
11 halt at end of current game

```

The program itself occupies locations 0-740. The table of moves, list of current moves, and 32 HI-Q boards occupy the remainder of memory, starting at 1000. The program contains a copy of the initial HI-Q board, with others generated at each move.



Internally, a HI-Q board occupies (100)<sub>8</sub> locations, stored as an 8 x 8 array. A particular location on a board is defined by its X and Y coordinates, each of which may be in the range 0 to 6. The coordinates of the board positions are interpreted as follows:

|   |   |   |   |   |   |   |   |
|---|---|---|---|---|---|---|---|
| X | 0 | 1 | 2 | 3 | 4 | 5 | 6 |
| Y |   |   |   |   |   |   |   |
| 0 |   |   | * | * | * |   |   |
| 1 |   |   | * | * | * |   |   |
| 2 | * | * | * | * | * | * | * |
| 3 | * | * | * | + | * | * | * |
| 4 | * | * | * | * | * | * | * |
| 5 |   |   | * | * | * |   |   |
| 6 |   |   | * | * | * |   |   |

The asterisks denote positions which initially contain a peg; the plus sign represents the initial empty hole. While a 7 x 7 array would suffice to store a board, the use of an 8 x 8 array makes interpretation of stored coordinates easier, as the X and Y coordinates are represented by individual octal digits.

Each array location contains a 1 if a peg is present at the corresponding board location, a 0 if a hole is present, or a -1 if the corresponding location is off the board (e.g. (X,Y)=(0,1) or (3,7)). The contents of the array are updated whenever a move is made.

The program starts at BEGIN (200). The current move indicator NMOVE is set to 1, the autoindex AXMOV is set to the beginning of free space to build the table of possible moves, and the game counter is set to 1.

SCAN begins the code to examine each location of the current board and build a table of possible moves. For each position, a check is made that that position contains a peg, then a check is made in each of the four possible directions to determine if a move is possible. When a possible move (the adjacent hole contains a peg and the next hole is empty) is found, an entry is made in the move table. This entry contains the Y and X coordinates of the peg which will move in the first two octal digits, and the Y and X coordinates of the location to which it will jump in the last two digits. After all board locations have been scanned for possible moves, a zero is stored in the move table to separate this move table from the one which will be built at the next level.



CHEQUE picks up the current-level move table pointer (MOV\_TBL contains a pointer to the current move at each level) and branches to EMPTY if no more moves remain to be made. Otherwise, NMOVE is incremented to indicate the number of the next move to be made, and the current board is copied into the next higher board location. The coordinates contained in the move-table entry are used, together with the address of the newly-created board, to "move" the selected peg and remove the peg which was jumped.

NMOVE is examined to determine whether 32 moves have now been made. If not, a branch is made to SCAN to generate the move table for the newly-created board. If so, the board may be printed, depending on the location of the final peg and the output print option selected.

EMPTY is reached when the move table at a particular level is empty. Depending on the options selected, the number of moves made or the current series of boards may be printed. Switch 11 is examined to determine whether the program should halt.

NMOVE is then decremented so that the program may back up one level. The NMOVEth entry in MOV\_TBL is found, and this is the address of the last move made at level NMOVE. This address is used to initialize a search for the end of the level NMOVE move table, which is in turn used to reset AXMOV to build the new move tables for succeeding levels. The pointer in MOV\_TBL is incremented to point to the next available move, and the game counter is incremented. A branch is then made to CHEQUE, and the process continues.

### Operation

Normal operation of this program consists of setting the program counter to 0200 with LOAD ADDR, selecting the desired switch options, and pressing START. A typical printout is included. This printout represents the first solution found by the program, and includes the serial number (in octal) and the 32 boards generated. In the printout, an asterisk (\*) represents a peg, while a plus (+) represents an empty hole.

Because the program backs up and tries again from the last successful move made, it takes a very long time for the moves made in the early part of the game to change. The program can be made to try a new move earlier than it ordinarily would with the following procedure:

1. use switch 11 to halt at the end of current game
2. using DEPOSIT, set location NMOVE to the level at which the new move is to be made
3. reset the switches and restart at location 0513.



It may also be interesting to play the game with a different initial board configuration. This may be modified in core before starting the program. If this is done, it may be desirable either to alter the test for the desired final board configuration (now made at 464), or to operate the program so that it prints all one-peg games (switch 0 on).

The following is the first solution found by the HI-Q program.

00047425

```

*** *** *** +** *++ *++ *++ *+*
*** *+* *+* ++* ++* ++* ++* +++
***** ****+*** *++**** *+***** *+***** *++**** ++*+**** ++*+****
+* ****+*** ****+*** ****+*** ****+*** ****+*** ****+*** ****+***
***** ****+*** ****+*** ****+*** ****+*** ****+*** ****+*** ****+***
*?/? *** *** *** *** *** *** ***
/? * *** *** *** *** *** ***

+ *+* ++* ++* ++* ++* ++* ++*
+++ *++ +++ +++ *++ *++ *++ *++
++*+*++ ++++*++ ++*+*++ ++*+*++ ++++*++ ++++*++ ++++*++ ++++*++
***** *+***** *+***** ++*+***** ++*+***** ++*+***** ++*+***** ++*+*****
***** ****+*** ****+*** *+***** *+***** *+***** *+***** *+*****
*** *** *** *** *** *** *** ***
*** *** *** *** *** *** *** ***

+++ +++ +++ +++ +++ +++ +++ +++
*++ *++ +++ +++ +++ +++ +++ +++
++++*++ ++*+*++ ++++*++ ++++*++ ++++*++ ++++*++ ++++*++ ++++*++
++*+*++ ++++*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++
***** *+***** *+***** ++*+***** ++*+***** ++*+***** ++*+***** ++*+*****
*** *** *** *** *** *** *** ***
*** *** *** *** *** *** *** ***

+++ +++ +++ +++ +++ +++ +++ +++
+++ +++ +++ +++ +++ +++ +++ +++
++++*++ ++++*++ ++++*++ ++++*++ ++++*++ ++++*++ ++++*++ ++++*++
++*+*++ ++++*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++
++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++ ++*+*++
**+ **+ ++* ++* ++* ++* ++* ++*
+** *++ +++ +++ +++ +++ +++ +++

```

## / SWITCH FUNCTIONS...

```

/ 0 PRINT ALL ONE-PEG GAMES
/ 1 PRINT ALL GAMES
/ 2 PRINT (OCTAL) MOVE COUNTS FOR ALL GAMES
/ 10 PRINT ONLY GAME NUMBER, NOT BOARDS
/ 11 HALT AT END OF CURRENT GAME

```

```

0010 *10
0010 0000 AXMOV 0 STORE POINTER FOR MOVE TABLE
0011 0000 AXOCT 0

0020 *20
0020 0740 PBOARD BOARD-100
0021 1040 PBRD2 BOARD
0022 0000 NMOVE 0 MOVE COUNTER
0023 0777 PMOVTBL MOVTBL-1 POINTER TO MOVE TABLE POINTERS
0024 5037 PFREEESP FREESP-1
0025 0000 CURMOV 0 POINTER TO CURRENT ENTRY IN MOVTBL
0026 0000 X 0 CURRENT X
0027 0000 Y 0 Y
0030 0000 THIS 0
0031 0000 POINT 0

0032 0000 ENTER 0 PUT ENTRY IN MOVE TABLE
0033 1027 TAD Y PACK CURRENT Y
0034 4046 JMS R3L
0035 1026 TAD X CURRENT X
0036 4046 JMS R3L
0037 1027 TAD Y CURRENT Y AGAIN
0040 4046 JMS R3L
0041 1026 TAD X CURRENT X AGAIN
0042 1432 TAD I ENTER DISPLACEMENT
0043 3410 DCA I AXMOV STORE
0044 2032 ISZ ENTER
0045 5432 JMP I ENTER

0046 0000 R3L 0 ROTATE 3 LEFT
0047 7106 CLL RTL
0050 7004 RAL
0051 5446 JMP I R3L

0052 0000 TYPE 0
0053 6046 TLS
0054 6041 TSF
0055 5054 JMP .-1
0056 5452 JMP I TYPE

0057 0000 CRLF 0
0060 7200 CLA
0061 1067 TAD C215
0062 4052 JMS TYPE
0063 1070 TAD M3
0064 4052 JMS TYPE
0065 7200 CLA
0066 5457 JMP I CRLF
0067 0215 C215 215
0070 7775 M3 -3

```



|      |      |        |              |                               |
|------|------|--------|--------------|-------------------------------|
| 0071 | 0000 | SWITCH | 0            | READ SWITCH REGISTER AND MASK |
| 0072 | 7604 |        | CLA OSR      |                               |
| 0073 | 0471 |        | AND I SWITCH |                               |
| 0074 | 2071 |        | ISZ SWITCH   |                               |
| 0075 | 5471 |        | JMP I SWITCH |                               |
|      |      |        |              |                               |
| 0076 | 0000 | OCTAL  | 0            |                               |
| 0077 | 7104 |        | CLL RAL      |                               |
| 0100 | 3117 |        | DCA OCT1     |                               |
| 0101 | 1121 |        | TAD OCM4     |                               |
| 0102 | 3120 |        | DCA OCT2     |                               |
| 0103 | 1117 | OCT0   | TAD OCT1     |                               |
| 0104 | 7006 |        | RTL          |                               |
| 0105 | 7004 |        | RAL          |                               |
| 0106 | 3117 |        | DCA OCT1     |                               |
| 0107 | 1117 |        | TAD OCT1     |                               |
| 0110 | 0122 |        | AND OCT7     |                               |
| 0111 | 1123 |        | TAD OCT260   |                               |
| 0112 | 4052 |        | JMS TYPE     |                               |
| 0113 | 7200 |        | CLA          |                               |
| 0114 | 2120 |        | ISZ OCT2     |                               |
| 0115 | 5103 |        | JMP OCT0     |                               |
| 0116 | 5476 |        | JMP I OCTAL  |                               |
| 0117 | 0000 | OCT1   | 0            |                               |
| 0120 | 0000 | OCT2   | 0            |                               |
| 0121 | 7774 | OCM4   | -4           |                               |
| 0122 | 0007 | OCT7   | 7            |                               |
| 0123 | 0260 | OCT260 | 260          |                               |
|      |      |        |              |                               |
| 0124 | 0000 | XINIT  | 0            | INITIALIZE GAME COUNTER       |
| 0125 | 7200 |        | CLA          |                               |
| 0126 | 1134 |        | TAD XM10     |                               |
| 0127 | 3133 |        | DCA XCOUNT   |                               |
| 0130 | 1135 |        | TAD PXBUF    |                               |
| 0131 | 3011 |        | DCA AXOCT    |                               |
| 0132 | 5524 |        | JMP I XINIT  |                               |
| 0133 | 0000 | XCOUNT | 0            |                               |
| 0134 | 7766 | XM10   | -10.         |                               |
| 0135 | 0135 | PXBUF  | XBUF-1       |                               |
| 0136 |      | XBUF   | BSS 10.      |                               |



```

0200 *200
/***** BEGIN HERE TO START FROM FIRST POSSIBLE GAME
0200 7201 BEGIN CLA IAC SET MOVE=1
0201 3022 DCA NMOVE
0202 1024 TAD PFREESP SETT UP STORE POINTER TO FREE AREA
0203 3010 DCA AXMOV
0204 4124 JMS XINIT SET GAME COUNT TO 1
0205 3411 DCA I AXOCT
0206 2133 ISZ XCOUNT
0207 5205 JMP .-2
0210 2147 ISZ XBUF+9.

0211 1022 SCAN TAD NMOVE SCAN CURRENT BOARD AND BUILD MOVE TABI
0212 1023 TAD PMOVTBL
0213 3025 DCA CURMOV
0214 1010 TAD AXMOV NEXT FRREE SPACE
0215 7001 IAC
0216 3425 DCA I CURMOV
0217 7326 P2AC SET Y,X TO 0,2
0220 3026 DCA X
0221 3027 DCA Y

0222 1022 SCN00 TAD NMOVE COMPUTE ADDR OF THIS BOARD
0223 4046 JMS R3L
0224 1027 TAD Y
0225 4046 JMS R3L
0226 1026 TAD X
0227 1020 TAD PBOARD
0230 3030 DCA THIS ADDR OF CURRENT CONSIDERED POINT

0231 1430 TAD I THIS CHECK FOR PEG PRESENT
0232 7750 SPA SNA CLA
0233 5340 JMP SCN10 NO

0234 1026 TAD X CHECK X LE 4
0235 1356 TAD M4
0236 7740 SMA SZA CLA
0237 5254 JMP SCN02 NO

0240 1030 TAD THIS CHECK N(Y,X+1) G 0
0241 7001 IAC
0242 3357 DCA TEM
0243 1757 TAD I TEM
0244 7750 SPA SNA CLA
0245 5254 JMP SCN02 NO

0246 2357 ISZ TEM CHECK N(Y,X+2) ZERO
0247 1757 TAD I TEM
0250 7640 SZA CLA
0251 5254 JMP SCN02 NO
0252 4032 JMS ENTER YES, STORE POINT IN TABLE
0253 0002 2

```

|      |      |       |             |                         |
|------|------|-------|-------------|-------------------------|
| 0254 | 1027 | SCN02 | TAD Y       | IS Y LE 4               |
| 0255 | 1356 |       | TAD M4      |                         |
| 0256 | 7740 |       | SMA SZA CLA |                         |
| 0257 | 5276 |       | JMP SCN04   | NO                      |
| 0260 | 1030 |       | TAD THIS    | CHECK N(Y+1,X) POS      |
| 0261 | 1360 |       | TAD C10     |                         |
| 0262 | 3357 |       | DCA TEM     |                         |
| 0263 | 1757 |       | TAD I TEM   |                         |
| 0264 | 7750 |       | SPA SNA CLA |                         |
| 0265 | 5276 |       | JMP SCN04   | NO                      |
| 0266 | 1357 |       | TAD TEM     | CHECK N(Y+2,X) EMPTY    |
| 0267 | 1360 |       | TAD C10     |                         |
| 0270 | 3357 |       | DCA TEM     |                         |
| 0271 | 1757 |       | TAD I TEM   |                         |
| 0272 | 7640 |       | SZA CLA     |                         |
| 0273 | 5276 |       | JMP SCN04   | NO                      |
| 0274 | 4032 |       | JMS ENTER   | YES, STORE POINT        |
| 0275 | 0020 |       | 20          |                         |
| 0276 | 7344 | SCN04 | M2AC        | CHECK X GE 2            |
| 0277 | 1026 |       | TAD X       |                         |
| 0300 | 7710 |       | SPA CLA     |                         |
| 0301 | 5316 |       | JMP SCN06   | NO                      |
| 0302 | 7344 |       | M2AC        |                         |
| 0303 | 1030 |       | TAD THIS    |                         |
| 0304 | 3357 |       | DCA TEM     |                         |
| 0305 | 1757 |       | TAD I TEM   |                         |
| 0306 | 7640 |       | SZA CLA     |                         |
| 0307 | 5316 |       | JMP SCN06   | NO                      |
| 0310 | 2357 |       | ISZ TEM     | CHECK N(Y,X-1) POSITIVE |
| 0311 | 1757 |       | TAD I TEM   |                         |
| 0312 | 7750 |       | SPA SNA CLA |                         |
| 0313 | 5316 |       | JMP SCN06   |                         |
| 0314 | 4032 |       | JMS ENTER   |                         |
| 0315 | 7776 |       | -2          |                         |



|      |      |       |             |                                     |
|------|------|-------|-------------|-------------------------------------|
| 0316 | 7344 | SCN06 | M2AC        | CHECK Y GE 2                        |
| 0317 | 1027 |       | TAD Y       |                                     |
| 0320 | 7710 |       | SPA CLA     |                                     |
| 0321 | 5340 |       | JMP SCN10   |                                     |
| 0322 | 1030 |       | TAD THIS    | N(Y-1,X) POS                        |
| 0323 | 1361 |       | TAD M10     |                                     |
| 0324 | 3357 |       | DCA TEM     |                                     |
| 0325 | 1757 |       | TAD I TEM   |                                     |
| 0326 | 7750 |       | SPA SNA CLA |                                     |
| 0327 | 5340 |       | JMP SCN10   | NO                                  |
| 0330 | 1357 |       | TAD TEM     | N(Y-2,X) ZERO                       |
| 0331 | 1361 |       | TAD M10     |                                     |
| 0332 | 3357 |       | DCA TEM     |                                     |
| 0333 | 1757 |       | TAD I TEM   |                                     |
| 0334 | 7640 |       | SZA CLA     |                                     |
| 0335 | 5340 |       | JMP SCN10   |                                     |
| 0336 | 4032 |       | JMS ENTER   |                                     |
| 0337 | 7760 |       | -20         |                                     |
| 0340 | 2026 | SCN10 | ISZ X       | NEXT X VALUE                        |
| 0341 | 1026 |       | TAD X       | X GE 7                              |
| 0342 | 1362 |       | TAD M7      |                                     |
| 0343 | 7710 |       | SPA CLA     | SKIP IF SO                          |
| 0344 | 5222 |       | JMP SCN00   | NO, SCAN NEXT POINT                 |
| 0345 | 3026 |       | DCA X       | RESET X TO 0                        |
| 0346 | 2027 |       | ISZ Y       | NEXT Y                              |
| 0347 | 1027 |       | TAD Y       | CHECK                               |
| 0350 | 1362 |       | TAD M7      | GE 7                                |
| 0351 | 7710 |       | SPA CLA     | SKIP IF SO                          |
| 0352 | 5222 |       | JMP SCN00   | NO                                  |
| 0353 | 3410 |       | DCA I AXMOV | STORE END-OF-MOVE CHAR AND END SCAN |
| 0354 | 5755 |       | JMP I .+1   |                                     |
| 0355 | 0400 |       | CHEQUE      |                                     |
| 0356 | 7774 | M4    | -4          |                                     |
| 0357 | 0000 | TEM   | 0           |                                     |
| 0360 | 0010 | C10   | 10          |                                     |
| 0361 | 7770 | M10   | -10         |                                     |
| 0362 | 7771 | M7    | -7          |                                     |

|      |      |                     |                                       |
|------|------|---------------------|---------------------------------------|
| 0400 |      | *400                |                                       |
| 0400 | 1425 | CHEQUE TAD I CURMOV | CONTENTS OF CURRENT MOVE POINTER      |
| 0401 | 3031 | DCA POINT           | POINTS TO CURRENT ENTRY               |
| 0402 | 1431 | TAD I POINT         | MORE MOVES REMAINING                  |
| 0403 | 7650 | SNA CLA             |                                       |
| 0404 | 5272 | JMP EMPTY           | NO                                    |
| 0405 | 1022 | TAD NMOVE           | COMPUTE ADDR OF THIS BOARD            |
| 0406 | 4046 | JMS R3L             |                                       |
| 0407 | 4046 | JMS R3L             |                                       |
| 0410 | 1020 | TAD PBOARD          |                                       |
| 0411 | 3347 | DCA TEM1            |                                       |
| 0412 | 2022 | ISZ NMOVE           | INCR MOVES AND COMPUTE ADDR OF NEXT B |
| 0413 | 1022 | TAD NMOVE           |                                       |
| 0414 | 4046 | JMS R3L             |                                       |
| 0415 | 4046 | JMS R3L             |                                       |
| 0416 | 1020 | TAD PBOARD          |                                       |
| 0417 | 3350 | DCA TEM2            |                                       |
| 0420 | 1353 | TAD M100            | SET UP WORD COUNT                     |
| 0421 | 3351 | DCA TEM3            |                                       |
| 0422 | 1747 | TAD I TEM1          | MOVE                                  |
| 0423 | 3750 | DCA I TEM2          |                                       |
| 0424 | 2347 | ISZ TEM1            |                                       |
| 0425 | 2350 | ISZ TEM2            |                                       |
| 0426 | 2351 | ISZ TEM3            | COUNT                                 |
| 0427 | 5222 | JMP .-5             |                                       |
| 0430 | 1431 | TAD I POINT         | GET CURRENT MOVE                      |
| 0431 | 7112 | CLL RTR             | FROM                                  |
| 0432 | 7012 | RTR                 |                                       |
| 0433 | 7012 | RTR                 |                                       |
| 0434 | 0354 | AND C77             |                                       |
| 0435 | 1347 | TAD TEM1            | ADDR OF CURRENT ARRAY                 |
| 0436 | 3350 | DCA TEM2            |                                       |
| 0437 | 3750 | DCA I TEM2          | REMOVE THE PEG                        |
| 0440 | 1431 | TAD I POINT         |                                       |
| 0441 | 0354 | AND C77             | TO                                    |
| 0442 | 1347 | TAD TEM1            |                                       |
| 0443 | 3351 | DCA TEM3            |                                       |
| 0444 | 7001 | IAC                 | PUT A PEG THERE                       |
| 0445 | 3751 | DCA I TEM3          |                                       |
| 0446 | 7100 | CLL                 | REMOVE JUMPED PAG                     |
| 0447 | 1350 | TAD TEM2            |                                       |
| 0450 | 1351 | TAD TEM3            |                                       |
| 0451 | 7010 | RAR                 |                                       |
| 0452 | 3351 | DCA TEM3            |                                       |
| 0453 | 3751 | DCA I TEM3          |                                       |



|                                         |      |        |              |                                      |
|-----------------------------------------|------|--------|--------------|--------------------------------------|
| 0454                                    | 1022 |        | TAD NMOVE    | 32 MOVES MADE                        |
| 0455                                    | 1352 |        | TAD M32D     |                                      |
| 0456                                    | 7640 |        | SZA CLA      | SKIP IF SO, OTHERWISE                |
| 0457                                    | 5755 |        | JMP I SCANP  | GO MAKE NEXT MOVE                    |
| 0460                                    | 4071 |        | JMS SWITCH   |                                      |
| 0461                                    | 4000 |        | 4000         | SHOULD ALL ONE-PEG GAMES BE PRINTED  |
| 0462                                    | 7640 |        | SZA CLA      | SKIP IF NOT                          |
| 0463                                    | 5267 |        | JMP .+4      | BRANCH IF SO                         |
| 0464                                    | 1760 |        | TAD I LMOVP  | LAST MOVE CORRECT                    |
| 0465                                    | 7650 |        | SNA CLA      |                                      |
| 0466                                    | 5272 |        | JMP EMPTY    | NO                                   |
| 0467                                    | 1022 |        | TAD NMOVE    | PRINT                                |
| 0470                                    | 4756 |        | JMS I OUTBDP |                                      |
| 0471                                    | 5307 |        | JMP EMPTY2   | PROCEED TO NEXT GAME                 |
| 0472                                    | 4071 | EMPTY  | JMS SWITCH   |                                      |
| 0473                                    | 1000 |        | 1000         | END OF GAME..SHOULD NMOVE BE PRINTED |
| 0474                                    | 7650 |        | SNA CLA      | SKIP IF SO                           |
| 0475                                    | 5301 |        | JMP .+4      | BRANCH IF NOT                        |
| 0476                                    | 1022 |        | TAD NMOVE    |                                      |
| 0477                                    | 4076 |        | JMS OCTAL    |                                      |
| 0500                                    | 4057 |        | JMS CRLF     |                                      |
| 0501                                    | 4071 |        | JMS SWITCH   |                                      |
| 0502                                    | 2000 |        | 2000         | SHOULD ALL GAMES BE PRINTED          |
| 0503                                    | 7650 |        | SNA CLA      | SKIP IF SO                           |
| 0504                                    | 5307 |        | JMP .+3      | BRANCH IF NOT                        |
| 0505                                    | 1022 |        | TAD NMOVE    | PRINT                                |
| 0506                                    | 4756 |        | JMS I OUTBDP |                                      |
| 0507                                    | 4071 | EMPTY2 | JMS SWITCH   |                                      |
| 0510                                    | 0001 |        | 0001         | HALT AT END OF GAME                  |
| 0511                                    | 7640 |        | SZA CLA      | SKIP IF NO                           |
| 0512                                    | 7402 |        | HLT          | YES                                  |
| /***** BEGIN HERE TO RESTART AFTER HALT |      |        |              |                                      |
| 0513                                    | 7240 |        | STA          | DECREMENT MOVE COUNTER               |
| 0514                                    | 1022 |        | TAD NMOVE    |                                      |
| 0515                                    | 7450 |        | SNA          | ZERO                                 |
| 0516                                    | 5345 |        | JMP QUIT     | YES, ALL GAMES PLAYED                |
| 0517                                    | 3022 |        | DCA NMOVE    |                                      |
| 0520                                    | 1022 |        | TAD NMOVE    | COMPUTE ADDR OF MOVE TABLE ENTRIES   |
| 0521                                    | 1023 |        | TAD PMOVTBL  |                                      |
| 0522                                    | 3025 |        | DCA CURMOV   |                                      |
| 0523                                    | 7240 |        | STA          | RESET THE STORE POINTER              |
| 0524                                    | 1425 |        | TAD I CURMOV |                                      |
| 0525                                    | 3010 |        | DCA AXMOV    |                                      |
| 0526                                    | 1410 |        | TAD I AXMOV  |                                      |
| 0527                                    | 7640 |        | SZA CLA      | FIND END                             |
| 0530                                    | 5326 |        | JMP .-2      |                                      |
| 0531                                    | 2425 |        | ISZ I CURMOV | NEXT MOVE IN TABLE                   |

|      |      |        |                  |                                   |
|------|------|--------|------------------|-----------------------------------|
| 0532 | 4124 |        | JMS XINIT        | INCR GAME COUNT                   |
| 0533 | 1357 |        | TAD PXBUF2       | END OF BUFFER                     |
| 0534 | 3347 | INCR   | DCA TEM1         |                                   |
| 0535 | 2747 |        | ISZ I TEM1       |                                   |
| 0536 | 5200 |        | JMP CHEQUE       | JUMP IF NO CARRY                  |
| 0537 | 2133 |        | ISZ XCOUNT       | CONT CARRIES                      |
| 0540 | 7410 |        | SKP              |                                   |
| 0541 | 5200 |        | JMP CHEQUE       | OVERFLOW                          |
| 0542 | 7240 |        | STA              | MOVE TO NEXT WORD                 |
| 0543 | 1347 |        | TAD TEM1         |                                   |
| 0544 | 5334 |        | JMP INCR         |                                   |
|      |      |        |                  |                                   |
| 0545 | 7402 | QUIT   | HLT              |                                   |
| 0546 | 5345 |        | JMP .-1          |                                   |
|      |      |        |                  |                                   |
| 0547 | 0000 | TEM1   | 0                |                                   |
| 0550 | 0000 | TEM2   | 0                |                                   |
| 0551 | 0000 | TEM3   | 0                |                                   |
| 0552 | 7740 | M32D   | -32.             |                                   |
| 0553 | 7700 | M100   | -100             |                                   |
| 0554 | 0077 | C77    | 77               |                                   |
| 0555 | 0211 | SCANP  | SCAN             |                                   |
| 0556 | 0600 | OUTBDP | OUTBRD           |                                   |
| 0557 | 0147 | PXBUF2 | XBUF+9.          | LAST COUNT ENTRY                  |
| 0560 | 4773 | LMOVP  | 31.*100+BOARD+33 | ADDR OF CENTER HOLE OF LAST BOARD |



## / SUBROUTINE TO OUTPUT CURRENT SERIES OF BOARDS

```

0600 *600
0600 0000 OUTBRD 0
0601 3312 DCA OUTN
0602 4124 JMS XINIT OUTPUT GAME NUMBER
0603 1411 TAD I AXOCT
0604 7440 SZA DON'T PRINT ZERO WORDS
0605 5210 JMP .+3
0606 2133 ISZ XCOUNT
0607 5203 JMP .-4 GET NEXT WORD
0610 4076 JMS OCTAL PRINT NONZERO
0611 2133 ISZ XCOUNT COUNT
0612 7410 SKP
0613 5216 JMP .+3
0614 1411 TAD I AXOCT
0615 5210 JMP .-5
0616 4057 JMS CRLF
0617 4071 JMS SWITCH TEST 'PRINT OCTAL ONLY'
0620 0002 0002
0621 7640 SZA CLA
0622 5600 JMP I OUTBRD

0623 1021 TAD PBRD2
0624 3313 DCA OUTPTR
0625 1312 OUT00 TAD OUTN
0626 7750 SPA SNA CLA
0627 5600 JMP I OUTBRD
0630 1312 TAD OUTN
0631 1331 TAD OUTM8
0632 3312 DCA OUTN DECREMENT NUMBER BY 8
0633 1312 TAD OUTN
0634 7500 SMA IF MORE THAN 8 REMAIN
0635 7200 CLA SET NUMBER TO 8
0636 1326 TAD C8
0637 7041 CMA IAC NEGATE NUMBER
0640 3314 DCA OUTCNT STORE IT
0641 1330 OUT02 TAD OUTM7
0642 3315 DCA OUTLCT
0643 1314 OUT03 TAD OUTCNT
0644 3316 DCA OUTEMP
0645 1313 TAD OUTPTR
0646 3317 DCA OTEMP
0647 1330 OUT04 TAD OUTM7
0650 3320 DCA OUCHRC
0651 1717 OUT05 TAD I OTEMP
0652 7540 SMA SZA
0653 1321 TAD STARCH '0-1
0654 7450 SNA
0655 1322 TAD PLUSCH '+'
0656 7510 SPA
0657 1323 TAD BLNKCH ' +1
0660 4052 JMS TYPE
0661 7200 CLA
0662 2317 ISZ OTEMP
0663 2320 ISZ OUCHRC
0664 5251 JMP OUT05
0665 1324 TAD C240
0666 4052 JMS TYPE

```

|      |      |             |                          |
|------|------|-------------|--------------------------|
| 0667 | 4052 | JMS TYPE    |                          |
| 0670 | 7200 | CLA         |                          |
| 0671 | 1317 | TAD OTEMP   |                          |
| 0672 | 1325 | TAD 64M7    |                          |
| 0673 | 3317 | DCA OTEMP   |                          |
| 0674 | 2316 | ISZ OUTEMP  | COUNT BOARDS             |
| 0675 | 5247 | JMP OUT04   |                          |
| 0676 | 1313 | TAD OUTPTR  | MOVE TO NEXT ROW         |
| 0677 | 1326 | TAD C8      |                          |
| 0700 | 3313 | DCA OUTPTR  |                          |
| 0701 | 4057 | JMS CRLF    |                          |
| 0702 | 2315 | ISZ OUTLCT  | COUNT LINES              |
| 0703 | 5243 | JMP OUT03   |                          |
| 0704 | 4057 | JMS CRLF    |                          |
| 0705 | 4057 | JMS CRLF    |                          |
| 0706 | 4057 | JMS CRLF    |                          |
| 0707 | 1313 | TAD OUTPTR  | MOVE UP TO NEXT 8 BOARDS |
| 0710 | 1327 | TAD C710    |                          |
| 0711 | 5224 | JMP OUT00-1 |                          |
| 0712 | 0000 | OUTN        | 0                        |
| 0713 | 0000 | OUTPTR      | 0                        |
| 0714 | 0000 | OUTCNT      | 0                        |
| 0715 | 0000 | OUTLCT      | 0                        |
| 0716 | 0000 | OUTEMP      | 0                        |
| 0717 | 0000 | OTEMPT      | 0                        |
| 0720 | 0000 | OUCHRC      | 0                        |
| 0721 | 0251 | STARCH      | '*-1                     |
| 0722 | 0253 | PLUSCH      | '+                       |
| 0723 | 0241 | BLNKCH      | ' +1                     |
| 0724 | 0240 | C240        | '                        |
| 0725 | 0071 | 64M7        | 100-7                    |
| 0726 | 0010 | C8          | 8                        |
| 0727 | 0710 | C710        | 710                      |
| 0730 | 7771 | OUTM7       | -7                       |
| 0731 | 7770 | OUTM8       | -8.                      |

1000-7\*10



|      |        |          |
|------|--------|----------|
| 1000 |        | *1000    |
| 1000 | MOVTBL | BSS 32.  |
| 1040 | 7777   | BOARD -1 |
| 1041 | 7777   | -1       |
| 1042 | 0001   | 1        |
| 1043 | 0001   | 1        |
| 1044 | 0001   | 1        |
| 1045 | 7777   | -1       |
| 1046 | 7777   | -1       |
| 1047 | 7777   | -1       |
| 1050 | 7777   | -1       |
| 1051 | 7777   | -1       |
| 1052 | 0001   | 1        |
| 1053 | 0001   | 1        |
| 1054 | 0001   | 1        |
| 1055 | 7777   | -1       |
| 1056 | 7777   | -1       |
| 1057 | 7777   | -1       |
| 1060 | 0001   | 1        |
| 1061 | 0001   | 1        |
| 1062 | 0001   | 1        |
| 1063 | 0001   | 1        |
| 1064 | 0001   | 1        |
| 1065 | 0001   | 1        |
| 1066 | 0001   | 1        |
| 1067 | 7777   | -1       |
| 1070 | 0001   | 1        |
| 1071 | 0001   | 1        |
| 1072 | 0001   | 1        |
| 1073 | 0000   | 0        |
| 1074 | 0001   | 1        |
| 1075 | 0001   | 1        |
| 1076 | 0001   | 1        |
| 1077 | 7777   | -1       |
| 1100 | 0001   | 1        |
| 1101 | 0001   | 1        |
| 1102 | 0001   | 1        |
| 1103 | 0001   | 1        |
| 1104 | 0001   | 1        |
| 1105 | 0001   | 1        |
| 1106 | 0001   | 1        |
| 1107 | 7777   | -1       |
| 1110 | 7777   | -1       |
| 1111 | 7777   | -1       |
| 1112 | 0001   | 1        |
| 1113 | 0001   | 1        |
| 1114 | 0001   | 1        |
| 1115 | 7777   | -1       |
| 1116 | 7777   | -1       |
| 1117 | 7777   | -1       |
| 1120 | 7777   | -1       |
| 1121 | 7777   | -1       |
| 1122 | 0001   | 1        |
| 1123 | 0001   | 1        |
| 1124 | 0001   | 1        |
| 1125 | 7777   | -1       |
| 1126 | 7777   | -1       |
| 1127 | 7777   | -1       |

PAGE 12

'HI-Q' PROGRAM

SEPT 17 1970

|      |        |             |
|------|--------|-------------|
| 1130 | 7777   | -1          |
| 1131 | 7777   | -1          |
| 1132 | 7777   | -1          |
| 1133 | 7777   | -1          |
| 1134 | 7777   | -1          |
| 1135 | 7777   | -1          |
| 1136 | 7777   | -1          |
| 1137 | 7777   | -1          |
| 1140 |        | BSS 31.*100 |
| 5040 | FREESP | BSS 0       |



## SYMBOL TABLE

|        |      |        |      |
|--------|------|--------|------|
| AXMOV  | 0010 | PMOVTB | 0023 |
| AXOCT  | 0011 | POINT  | 0031 |
| BEGIN  | 0200 | PXBUF  | 0135 |
| BLNKCH | 0723 | PXBUF2 | 0557 |
| BOARD  | 1040 | QUIT   | 0545 |
| CHEQUE | 0400 | R3L    | 0046 |
| CRLF   | 0057 | SCAN   | 0211 |
| CURMOV | 0025 | SCANP  | 0555 |
| C10    | 0360 | SCN00  | 0222 |
| C215   | 0067 | SCN02  | 0254 |
| C240   | 0724 | SCN04  | 0276 |
| C710   | 0727 | SCN06  | 0316 |
| C77    | 0554 | SCN10  | 0340 |
| C8     | 0726 | STARCH | 0721 |
| EMPTY  | 0472 | SWITCH | 0071 |
| EMPTY2 | 0507 | TEM    | 0357 |
| ENTER  | 0032 | TEM1   | 0547 |
| FREESP | 5040 | TEM2   | 0550 |
| INCR   | 0534 | TEM3   | 0551 |
| LMOVP  | 0560 | THIS   | 0030 |
| MOVTBL | 1000 | TYPE   | 0052 |
| M10    | 0361 | X      | 0026 |
| M100   | 0553 | XBUF   | 0136 |
| M3     | 0070 | XCOUNT | 0133 |
| M32D   | 0552 | XINIT  | 0124 |
| M4     | 0356 | XM10   | 0134 |
| M7     | 0362 | Y      | 0027 |
| NMOVE  | 0022 | 64M7   | 0725 |
| OCM4   | 0121 |        |      |
| OCTAL  | 0076 |        |      |
| OCT0   | 0103 |        |      |
| OCT1   | 0117 |        |      |
| OCT2   | 0120 |        |      |
| OCT260 | 0123 |        |      |
| OCT7   | 0122 |        |      |
| OTEMPT | 0717 |        |      |
| OUCHRC | 0720 |        |      |
| OUTBDP | 0556 |        |      |
| OUTBRD | 0600 |        |      |
| OUTCNT | 0714 |        |      |
| OUTEMP | 0716 |        |      |
| OUTLCT | 0715 |        |      |
| OUTM7  | 0730 |        |      |
| OUTM8  | 0731 |        |      |
| OUTN   | 0712 |        |      |
| OUTPTR | 0713 |        |      |
| OUT00  | 0625 |        |      |
| OUT02  | 0641 |        |      |
| OUT03  | 0643 |        |      |
| OUT04  | 0647 |        |      |
| OUT05  | 0651 |        |      |
| PBOARD | 0020 |        |      |
| PBRD2  | 0021 |        |      |
| PFREES | 0024 |        |      |
| PLUSCH | 0722 |        |      |

NO ERRORS IN ABOVE ASSEMBLY

## REFERENCES TO DEFINED SYMBOLS

| SYMBOL | VALUE | REFERENCES                                        |
|--------|-------|---------------------------------------------------|
| AXMOV  | 0010  | 0043 0203 0214 0353 0525 0526                     |
| AXOCT  | 0011  | 0131 0205 0603 0614                               |
| BEGIN  | 0200  |                                                   |
| BLNKCH | 0723  | 0657                                              |
| BOARD  | 1040  | 0020 0021 0560                                    |
| CHEQUE | 0400  | 0355 0536 0541                                    |
| CRLF   | 0057  | 0066 0500 0616 0701 0704 0705 0706                |
| CURMOV | 0025  | 0213 0216 0400 0522 0524 0531                     |
| C10    | 0360  | 0261 0267                                         |
| C215   | 0067  | 0061                                              |
| C240   | 0724  | 0665                                              |
| C710   | 0727  | 0710                                              |
| C77    | 0554  | 0434 0441                                         |
| C8     | 0726  | 0636 0677                                         |
| EMPTY  | 0472  | 0404 0466                                         |
| EMPTY2 | 0507  | 0471                                              |
| ENTER  | 0032  | 0042 0044 0045 0252 0274 0314 0336                |
| FREESP | 5040  | 0024                                              |
| INCR   | 0534  | 0544                                              |
| LMOVP  | 0560  | 0464                                              |
| MOVTBL | 1000  | 0023                                              |
| M10    | 0361  | 0323 0331                                         |
| M100   | 0553  | 0420                                              |
| M3     | 0070  | 0063                                              |
| M32D   | 0552  | 0455                                              |
| M4     | 0356  | 0235 0255                                         |
| M7     | 0362  | 0342 0350                                         |
| NMOVE  | 0022  | 0201 0211 0222 0405 0412 0413 0454 0467 0476 0505 |
|        |       | 0514 0517 0520                                    |
| OCM4   | 0121  | 0101                                              |
| OCTAL  | 0076  | 0116 0477 0610                                    |
| OCT0   | 0103  | 0115                                              |
| OCT1   | 0117  | 0100 0103 0106 0107                               |
| OCT2   | 0120  | 0102 0114                                         |
| OCT260 | 0123  | 0111                                              |
| OCT7   | 0122  | 0110                                              |
| OTEMPT | 0717  | 0646 0651 0662 0671 0673                          |
| OUCHRC | 0720  | 0650 0663                                         |
| OUTBDP | 0556  | 0470 0506                                         |
| OUTBRD | 0600  | 0556 0622 0627                                    |
| OUTCNT | 0714  | 0640 0643                                         |
| OUTEMP | 0716  | 0644 0674                                         |
| OUTLCT | 0715  | 0642 0702                                         |
| OUTM7  | 0730  | 0641 0647                                         |
| OUTM8  | 0731  | 0631                                              |
| OUTN   | 0712  | 0601 0625 0630 0632 0633                          |
| OUTPTR | 0713  | 0624 0645 0676 0700 0707                          |
| OUT00  | 0625  | 0711                                              |
| OUT02  | 0641  |                                                   |
| OUT03  | 0643  | 0703                                              |
| OUT04  | 0647  | 0675                                              |
| OUT05  | 0651  | 0664                                              |
| PBOARD | 0020  | 0227 0410 0416                                    |
| PBRD2  | 0021  | 0623                                              |
| PFREES | 0024  | 0202                                              |



|        |      |      |      |      |      |      |      |      |      |      |      |  |
|--------|------|------|------|------|------|------|------|------|------|------|------|--|
| PLUSCH | 0722 | 0655 |      |      |      |      |      |      |      |      |      |  |
| PMOVTB | 0023 | 0212 | 0521 |      |      |      |      |      |      |      |      |  |
| POINT  | 0031 | 0401 | 0402 | 0430 | 0440 |      |      |      |      |      |      |  |
| PXBUF  | 0135 | 0130 |      |      |      |      |      |      |      |      |      |  |
| PXBUF2 | 0557 | 0533 |      |      |      |      |      |      |      |      |      |  |
| QUIT   | 0545 | 0516 |      |      |      |      |      |      |      |      |      |  |
| R3L    | 0046 | 0034 | 0036 | 0040 | 0051 | 0223 | 0225 | 0406 | 0407 | 0414 | 0415 |  |
| SCAN   | 0211 | 0555 |      |      |      |      |      |      |      |      |      |  |
| SCANP  | 0555 | 0457 |      |      |      |      |      |      |      |      |      |  |
| SCN00  | 0222 | 0344 | 0352 |      |      |      |      |      |      |      |      |  |
| SCN02  | 0254 | 0237 | 0245 | 0251 |      |      |      |      |      |      |      |  |
| SCN04  | 0276 | 0257 | 0265 | 0273 |      |      |      |      |      |      |      |  |
| SCN06  | 0316 | 0301 | 0307 | 0313 |      |      |      |      |      |      |      |  |
| SCN10  | 0340 | 0233 | 0321 | 0327 | 0335 |      |      |      |      |      |      |  |
| STARCH | 0721 | 0653 |      |      |      |      |      |      |      |      |      |  |
| SWITCH | 0071 | 0073 | 0074 | 0075 | 0460 | 0472 | 0501 | 0507 | 0617 |      |      |  |
| TEM    | 0357 | 0242 | 0243 | 0246 | 0247 | 0262 | 0263 | 0266 | 0270 | 0271 | 0304 |  |
|        |      | 0305 | 0310 | 0311 | 0324 | 0325 | 0330 | 0332 | 0333 |      |      |  |
| TEM1   | 0547 | 0411 | 0422 | 0424 | 0435 | 0442 | 0534 | 0535 | 0543 |      |      |  |
| TEM2   | 0550 | 0417 | 0423 | 0425 | 0436 | 0437 | 0447 |      |      |      |      |  |
| TEM3   | 0551 | 0421 | 0426 | 0443 | 0445 | 0450 | 0452 | 0453 |      |      |      |  |
| THIS   | 0030 | 0230 | 0231 | 0240 | 0260 | 0303 | 0322 |      |      |      |      |  |
| TYPE   | 0052 | 0056 | 0062 | 0064 | 0112 | 0660 | 0666 | 0667 |      |      |      |  |
| X      | 0026 | 0035 | 0041 | 0220 | 0226 | 0234 | 0277 | 0340 | 0341 | 0345 |      |  |
| XBUF   | 0136 | 0135 | 0210 | 0557 |      |      |      |      |      |      |      |  |
| XCOUNT | 0133 | 0127 | 0206 | 0537 | 0606 | 0611 |      |      |      |      |      |  |
| XINIT  | 0124 | 0132 | 0204 | 0532 | 0602 |      |      |      |      |      |      |  |
| XM10   | 0134 | 0126 |      |      |      |      |      |      |      |      |      |  |
| Y      | 0027 | 0033 | 0037 | 0221 | 0224 | 0254 | 0317 | 0346 | 0347 |      |      |  |
| 64M7   | 0725 | 0672 |      |      |      |      |      |      |      |      |      |  |

